
Apresentação 

Sobre este manual 

O intuito deste material é somente servir como um guia de utilização do sistema. Nele, você encontrará 
uma breve descrição da utilização de cada cadastro ou processo, dicas dos principais atributos, além de 
exemplos.  

Convenções Adotadas 

Um sinal de adição (+) entre duas teclas significa que você deve pressionar essas teclas 
simultaneamente. Por exemplo, “pressione ALT+TAB” significa que você deve manter a tecla ALT 
pressionada enquanto pressiona TAB. 

Os ícones a seguir identificam os diferentes tipos de texto que podem ser encontrados neste material:  
 

Ícone Indica que o texto é: 

 Uma observação importante 

 Uma dica ou sugestão útil 

 Um exemplo 

 Relativo aos principais cadastros/processos que dependem do 
cadastro citado. 

 Relativo ao nome do cadastro/processo no Banco de Dados. 

Manufatura 

Objetivo 

Mostrar o funcionamento do Processo Customizado existente no Mega Empresarial para a criação de 
Regra Específica para o Código Alternativo da Ordem. 

Nesse manual será possível identificar os pré-requisitos necessários para que o sistema possa ser 
executado usando uma regra específica e como criação e parametrizar essa regra específica. 

Parâmetros do Manufatura 

Dentro dos Parâmetros do Manufatura, pasta Ordem, foi criado um novo campo para identificar, em 
linhas gerais, se o usuário utiliza regra específica para a geração de código alternativo da Ordem. Caso 
esse parâmetro não seja marcado, o sistema continuará trabalhando da mesma forma que atua hoje em 
dia. 

A tela a seguir, mostra os Parâmetros do Manufatura com o novo campo: 

Global 

Processo Customizado 

O Processo Customizado disponibilizado pelo Mega Empresarial permite que o usuário possa inserir 
regras específicas para determinados pontos ou cálculos do sistema. 

O Processo que trata da regra específica para a geração do Código Alternativo da Ordem, está tabela 
MGGLO.GLO_PROCESSOCUSTOMIZADO com os seguintes valores: 

 Campo PRC_ST_NOMEPROCESSO: F_GERAALTERNATIVOORDEM; 

 Campo PRC_ST_OWNER: MGMAN; 

 Campo PRC_ST_PACKAGE: MAN_PCK_ORDEM; 

 

Para que esse Processo Customizado passe a ser executado durante a geração de uma Ordem, será 
necessário alterar os seguintes campos da tabela MGGLO.GLO_PROCESSOCUSTOMIZADO: 

 PRC_ST_NOMEOBJETO: deve conter o nome do usuário no banco e o nome objeto que o 
usuário criou contendo a regra específica (Exemplo: MGMAN.F_NOMEOBJETO); 

 PRC_BO_STATUS: deve ter o valor S; 

 

 A forma como o objeto deverá ser criada, será explicada no próximo item. 

 Não existe uma tela dentro do Mega Empresarial onde esses dados possam ser manipulados. 

 A regra só será utilizada pelo Sistema se esses valores forem informados corretamente. 

Banco de Dados 

Criação da Função Específica 

A criação da função específica deve atender literalmente ao que o sistema pede, pois o sucesso na 
execução da rotina dependerá da correta forma em que ela for construída. 

A função que gerará o Código Alternativo da Ordem deverá ser criada com as mesmas características 
existentes no exemplo que irá junto com essa funcionalidade. 

O mesmo exemplo está copiado logo abaixo e a explicação dos pontos mais importantes virá em 
seguida. 

 

Exemplo de Função 

 A função abaixo deve ser criada no Owner MGMAN; 

 

CREATE OR REPLACE FUNCTION F_GERAALTERNATIVOORDEM(pParam Varchar2) Return 

sys_refcursor Is 

 /* ESSA ROTINA É APENAS UM EXEMPLO DE USO PARA GERAÇÃO DO CÓDIGO ALTERNATIVO DA ORDEM 

*/ 

 

 /* Origem -> Ponto de origem do sistema que chamou essa função 

       P = Pedido de Venda (Fabricação Sob Encomenda) 

       M = MRP 

       O = Ordem (Manual) 

 */ 

 

 Result  Varchar2(50) := ''; 

 cAuxCur sys_refcursor; 

 vXML    xmltype; 

 -- Variáveis internas que armazenam os valores passados no parâmetro (pParam) 

 v_Org_Codigo     mgman.pro_ordens.org_in_codigo%type; 

 v_Fil_Codigo     mgman.pro_ordens.fil_in_codigo%type; 

 v_Ord_Seq_Codigo mgman.pro_ordens.ord_seq_in_codigo%type; 

 v_Ord_Codigo     mgman.pro_ordens.ord_in_codigo%type; 

 v_Origem         Char(1); 

 -- Variáveis para receber valores dos selects .. into 

 vPedido mgman.pro_origem.ped_in_codigo%type; 

 vSeqMRP mgman.pro_ordens.mrp_in_sequencia%type; 

BEGIN 

  -- Extraindo XML dos parâmetros 

  vXML             := xmltype.createXML(Upper(pParam)); 

  v_Org_Codigo     := MGGLO.GLO_PCK_XML.ExtractXmlToNumber(vXML,'P_ORG_CODIGO'    ); 

  v_Fil_Codigo     := MGGLO.GLO_PCK_XML.ExtractXmlToNumber(vXML,'P_FIL_CODIGO'    ); 

  v_Ord_Seq_Codigo := MGGLO.GLO_PCK_XML.ExtractXmlToNumber(vXML,'P_ORD_SEQ_CODIGO'); 

  v_Ord_Codigo     := MGGLO.GLO_PCK_XML.ExtractXmlToNumber(vXML,'P_ORD_CODIGO'    ); 

  v_Origem         := MGGLO.GLO_PCK_XML.ExtractXmlToString(vXML,'P_ORIGEM'        ); 

  -- Regra para Ordens geradas a partir de um Pedido Especial 

  If v_Origem = 'P' Then 

    Begin 

      Select O.Ped_In_Codigo 

        Into vPedido 

        From Mgman.Pro_Origem O 

       Where O.Org_Tab_In_Codigo = 53 

         And O.Org_Pad_In_Codigo = 1 

         And O.Org_In_Codigo     = v_Org_Codigo 

         And O.Org_Tau_St_Codigo = 'G' 

         And O.Ord_Tab_In_Codigo = 218 

         And O.Ord_Seq_In_Codigo = v_Ord_Seq_Codigo 

         And O.Ord_In_Codigo     = v_Ord_Codigo; 

    Exception When No_Data_Found Then 

      vPedido := 0; 

    End; 

    -- Monta a regra do Pedido Especial, caso encontre o Pedido (Nº Pedido + Ano) 

    If vPedido <> 0 Then 

      Result := to_char(vPedido)||' - '||to_char(Sysdate,'yyyy'); 

    End If; 

 

  -- Regra para Ordens geradas manualmente a partir da tela de Ordens 

  Elsif v_Origem = 'O' Then 

    If v_Ord_Codigo Is Not Null And v_Ord_Codigo <> 0 Then 

      Result := to_char(v_Ord_Codigo)||' - '||to_char(Sysdate,'yyyy'); 

    End If; 

 

  -- Regra para Ordens geradas pelo MRP 

  Elsif v_Origem = 'M' Then 

    Begin 

      Select O.Mrp_In_Sequencia 

        Into vSeqMRP 

        From Mgman.Pro_Ordens O 

       Where O.Org_Tab_In_Codigo = 53 

         And O.Org_Pad_In_Codigo = 1 

         And O.Org_In_Codigo     = v_Org_Codigo 

         And O.Org_Tau_St_Codigo = 'G' 

         And O.Ord_Tab_In_Codigo = 218 

         And O.Fil_In_Codigo     = v_Fil_Codigo 

         And O.Ord_Seq_In_Codigo = v_Ord_Seq_Codigo 

         And O.Ord_In_Codigo     = v_Ord_Codigo; 

     Exception When No_Data_Found Then 

       vSeqMRP := 0; 

     End; 

     -- Monta a regra do MRP, caso encontre a Sequência da Rodada (Nº Rodada + Ano) 

     If vSeqMRP <> 0 Then 

       Result := to_char(v_Ord_Codigo)||' - '||to_char(vSeqMRP)||' - 

'||to_char(Sysdate,'yyyy'); 

     End If; 

  End If; 

  -- Retorno com o valor calculado 

  open cAuxCur for select Result from Dual; 

  return cAuxcur; 

END; 

 

Explicação dos pontos importantes da função 

Dentre as questões mais importantes a serem utilizadas para a criação da função estão: 

 Sempre deverá uma Função (Function), ou seja, não poderá ser um procedimento (Procedure); 

 A função a ser criada sempre deverá receber um parâmetro do tipo Varchar2; 

 O retorno da função sempre deverá ser “sys_refcursor”; 

 Deverá ser criada uma varíavel do tipo “sys_refcursor”; 

 Deverá ser criada uma variável do tipo “xmltype”; 

 A variável do tipo “xmltype” deverá ser utilizada conforme exemplo, dentro do bloco “Extraindo 
XML dos parâmetros”; 

 O parâmetro pParam conterá os seguintes valores que poderão ser utilizados dentro da rotina: 

o P_ORG_CODIGO: Código da Organização no Mega Empresarial; 

o P_FIL_CODIGO: Código da Filial no Mega Empresarial; 

o P_ORD_SEQ_CODIGO: Sequência da Ordem nessa Filial do Mega Empresarial; 

o P_ORD_CODIGO: Código da Ordem; 

o P_ORIGEM: Ponto de origem da inclusão da Ordem no sistema: 

  P: Pedido de Venda Especial; 

 M: MRP; 

 O: Ordem; 

 No final da função, o comando “open VARIAVEL_DO_TIPO_SYS_REFCURSOR for select 
CALCULO_DO_CODIGO_ALTERNATIVO from Dual;” deverá ser colocado obrigatoriamente 
antes de criar a linha do retorno da função; 

 O retorno da função obrigatoriamente deverá ser a variável do tipo sys_refcursor. 

 

 Os parâmetros permitem uma grande flexibilidade para montar as regras do Código Alternativo da 
Ordem. Só pelo fato de existir a Origem da geração da Ordem é possível criar regras para cada 
um desses processos, por exemplo, Ordem geradas pelo Pedido Especial podem ter o Código do 
Pedido no campo Alternativo e assim por diante. 


